

Documentação – Nível 2

Sistema de Rotas e Navegação entre Cidades em Prolog

1. Introdução

Este módulo implementa um sistema de navegação entre cidades utilizando Prolog. A solução permite:

- Registrar estradas entre cidades com distâncias.
- Tratar conexões bidirecionais.
- Encontrar rotas completas entre duas cidades.
- Calcular a distância total percorrida.
- Evitar loops através da lista de visitados.
- Validar rotas com limite máximo de cidades.

O objetivo é exercitar lógica recursiva, manipulação de listas e construção de regras.

2. Base de Conhecimento (Fatos)

```
estrada(sao_paulo, campinas, 90).  
estrada(campinas, ribeirao_preto, 230).  
estrada(sao_paulo, santos, 70).  
estrada(sao_paulo, sao_jose_dos_campos, 80).  
estrada(sao_jose_dos_campos, taubate, 40).  
estrada(taubate, ubatuba, 50).  
estrada(santos, ubatuba, 200).  
estrada(ribeirao_preto, franca, 90).
```

Cada fato segue o formato:

```
estrada(CidadeA, CidadeB, Distância).
```

3. Regras do Sistema

3.1 Conexão Bidirecional

```
conectado(X, Y, Dist) :- estrada(X, Y, Dist).  
conectado(X, Y, Dist) :- estrada(Y, X, Dist).
```

Essa regra transforma a base de estradas em um grafo bidirecional.

3.2 Regra Principal de Rota

```
rota(Origem, Destino, Caminho, DistanciaTotal) :-  
    viajar(Origem, Destino, [Origem], Caminho, DistanciaTotal).
```

Inicia a navegação recursiva, adicionando a cidade inicial à lista de visitados.

3.3 Navegação Recursiva

Caso Base

```
viajar(Destino, Destino, Visitados, Caminho, 0) :-  
    reverse(Visitados, Caminho).
```

Quando se atinge o destino, o caminho é retornado invertido.

Passo Recursivo

```
viajar(Atual, Destino, Visitados, Caminho, DistTotal) :-  
    conectado(Atual, Proxima, DistanciaTrecho),  
    \+ member(Proxima, Visitados),  
    viajar(Proxima, Destino, [Proxima|Visitados], Caminho, DistRestante),  
    DistTotal is DistanciaTrecho + DistRestante.
```

Processo:

1. Encontra cidades conectadas.
 2. Verifica se ainda não foram visitadas.
 3. Chama recursivamente adicionando à lista.
 4. Soma as distâncias retornadas.
-

3.4 Rota com Limite de Cidades

```
rota_curta(Origem, Destino, MaxCidades) :-  
    rota(Origem, Destino, Caminho, _),  
    length(Caminho, Tamanho),  
    Tamanho =< MaxCidades.
```

Verifica se a rota encontrada respeita o limite máximo de cidades.

4. Funcionamento Geral

Este módulo demonstra:

- Recursão em profundidade
- Prevenção de loops via lista de visitados
- Manipulação de listas (`member/2`, `reverse/2`, `length/2`)

- Construção de grafos simples em Prolog
-